Pensieve header: Computations in the ybax algebra using the Drinfel'd double (at $\epsilon^2=0$.
Continues ExpDoubleEpsilonSquare5@@.nb.

The double at $\epsilon^2 = 0$

Utilities

Canonical Form:

ybax

In[ ]:=
```
CF[sd_SeriesData] := MapAt[CF, sd, 3];
CF[ℰ_] := ExpandDenominator@
   ExpandNumerator@Together[Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^CF[x]];
```

The Kronecker δ:

ybax

In[ ]:=
```
Kδ /: Kδ_{i_,j_} := If[i === j, 1, 0];
```

Equality and multiplication of perturbed Gaussians; $\mathbb{E}[L, Q, P]$ stands for $e^{L+Q} P$:

ybax

In[ ]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

Zip and Bind

ybax

In[ ]:=
```
{t*, b*, y*, a*, x*, z*} = {τ, β, η, α, ξ, ζ};
{τ*, β*, η*, α*, ξ*, ζ*} = {t, b, y, a, x, z}; (u_i_)* := (u*)_i;
```

ybax

In[ ]:=
```
expand[sd_SeriesData] := MapAt[expand, sd, 3];
expand[ℰ_] := Expand[ℰ];
Zip_{}[P_] := P;
Zip_{ζ_,ζs___}[P_] := (expand[P // Zip_{ζs}] /. f_. ζ^d_. :→ ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

ybax

In[∘]:=

```
QZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
    qt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} Q, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → qt.(zs + ys)];
    Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    simp /@ 𝔼[L, Q2, Det[qt] ℯ^(-Q2) Zip_{ζs}[ℯ^(Q1) (P /. zrule)]] ];
QZip_{ζs_List} := QZip_{ζs,CF};
```

Upper to lower and lower to Upper:

ybax

In[∘]:=

```
U2l = {B_{i_}^(p_·) → ℯ^(-p b_i), B^(p_·) → ℯ^(-p b), T_{i_}^(p_·) → ℯ^(p t_i), T^(p_·) → ℯ^(p t), 𝒜_{i_}^(p_·) → ℯ^(p α_i), 𝒜^(p_·) → ℯ^(p α)};
l2U = {ℯ^(c_· b_{i_} + d_·) :→ B_i^(-c) ℯ^d, ℯ^(c_· b + d_·) :→ B^(-c) ℯ^d,
    ℯ^(c_· t_{i_} + d_·) :→ T_i^c ℯ^d, ℯ^(c_· t + d_·) :→ T^c ℯ^d,
    ℯ^(c_· α_{i_} + d_·) :→ 𝒜_i^c ℯ^d, ℯ^(c_· α + d_·) :→ 𝒜^c ℯ^d,
    ℯ^(ℰ_) :→ ℯ^(Expand@ℰ)};
```

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard all of $P e^{Q}$ as a single"P". Here the z's are b and α and the ζ's are β and a.

ybax

In[∘]:=

```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
    zs = Table[ζ*, {ζ, ζs}];
    c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
    ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
    ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
    lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
    zrule = Thread[zs → lt.(zs + ys)];
    L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
    Q2 = (Q1 = Q /. U2l /. zrule) /. Alternatives @@ zs → 0;
    simp /@ 𝔼[L2, Q2, Det[lt] ℯ^(-L2-Q2) Zip_{ζs}[ℯ^(L1+Q1) (P /. U2l /. zrule)]] //. l2U ];
LZip_{ζs_List} := LZip_{ζs,CF};
```

ybax

In[∘]:=

```
Bind_{{}}[L_, R_] := L R;
Bind_{{is__}}[L_𝔼, R_𝔼] := Module[{n},
    Times[
        L /. Table[(v : b | B | t | T | a | x | y)_i → v_{n@i}, {i, {is}}],
        R /. Table[(v : β | τ | α | 𝒜 | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Flatten@Table[{β_{n@i},τ_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{x_{n@i},η_{n@i}},{i,{is}}]} ];
B_{l_List}[L_, R_] := Bind_l[L, R]; B_{is___}[L_, R_] := Bind_{{is}}[L, R];
```

The two halves

ybax

In[∘]:=

```
(*Hopf algebra on the a,x side*)
am[i_,j_->k_] := 𝔼[(α[i] + α[j]) a[k], (ⅇ^(-α[j]) ξ[i] + ξ[j]) x[k], 1 + O[ϵ]^2]
aΔ[i_->j_,k_] := 𝔼[α[i] (a[j] + a[k]), ξ[i] (x[j] + x[k]), 1 + ϵ ξ[i] x[k] (-a[j] + 1/2 ξ[i] x[j]) + O[ϵ]^2]
aS[i_] := 𝔼[-α[i] a[i], -ⅇ^α[i] ξ[i] x[i], 1 - ϵ ⅇ^α[i] ξ[i] x[i] (a[i] + 1/2 ⅇ^α[i] ξ[i] x[i]) + O[ϵ]^2]
aSi[i_] := 𝔼[-α[i] a[i], -ⅇ^α[i] ξ[i] x[i], 1 - ϵ ⅇ^α[i] ξ[i] x[i] (a[i] - 1 + 1/2 ⅇ^α[i] ξ[i] x[i]) + O[ϵ]^2]
(*Hopf algebra on the y,b side*)
bm[i_,j_→k_] := 𝔼[(β[i] + β[j]) b[k], (η[i] + η[j]) y[k], 1 - ϵ η[j] y[k] β[i] + O[ϵ]^2]
bΔ[i_->j_,k_] := 𝔼[β[i] (b[j] + b[k]), η[i] (ⅇ^(-b[k]) y[j] + y[k]), 1 + 1/2 ϵ η[i]^2 y[j] y[k] ⅇ^(-b[k]) + O[ϵ]^2]
bS[i_] := 𝔼[-β[i] b[i], -ⅇ^b[i] η[i] y[i], 1 - ϵ ⅇ^b[i] η[i] y[i] (β[i] + 1/2 ⅇ^b[i] η[i] y[i]) + O[ϵ]^2]
bSi[i_] := 𝔼[-β[i] b[i], -ⅇ^b[i] η[i] y[i], 1 - ϵ ⅇ^b[i] η[i] y[i] (β[i] - 1 + 1/2 ⅇ^b[i] η[i] y[i]) + O[ϵ]^2]
```

First check that on the generators this agrees with our conventions in SLPortfolio.pdf with $\hbar = \gamma = 1$

In[∘]:=

```
{
  "[a,x]" → ((𝔼[0, 0, a2 x1] ~B[1,2]~ am[1,2→1])[[3]] - (𝔼[0, 0, a1 x2] ~B[1,2]~ am[1,2→1])[[3]]),
  "[b,y]" → ((𝔼[0, 0, y2 b1] ~B[1,2]~ bm[1,2→1])[[3]] - (𝔼[0, 0, y1 b2] ~B[1,2]~ bm[1,2→1])[[3]])
 } /. {z_1 → z} // Simplify
(Δ[#] → Simplify@Normal@Last[𝔼[0, 0, #1] ~B1~ aΔ[1→1,2]]) & /@ {a, x}
(Δ[#] → Simplify@Normal@Last[𝔼[0, 0, #1] ~B1~ bΔ[1→1,2]]) & /@ {b, y}
{
  " S(a)  = " ((𝔼[0, 0, a1] ~B1~ aS1)[[3]]),
  " S(x)  = " ((𝔼[0, 0, x1] ~B1~ aS1)[[3]]),
  " S(b)  = " ((𝔼[0, 0, b1] ~B1~ bS1)[[3]]),
  " S(y)  = " → ((𝔼[0, 0, y1] ~B1~ bS1)[[3]])
 } /. {z_1 → z} // Simplify
```

Out[∘]= $\{[a,x] \to -x + O[\epsilon]^2,\ [b,y] \to -y\,\epsilon + O[\epsilon]^2\}$

Out[∘]= $\{\Delta[a] \to a_1 + a_2,\ \Delta[x] \to x_1 + (1 - \epsilon\, a_1)\, x_2\}$

Out[∘]= $\{\Delta[b] \to b_1 + b_2,\ \Delta[y] \to B_2\, y_1 + y_2\}$

Out[∘]= $\{-\ S(a)\ =\ a + O[\epsilon]^2,\ -\ S(x)\ =\ x -\ S(x)\ =\ a\,x\,\epsilon + O[\epsilon]^2,$
$-\ S(b)\ =\ b + O[\epsilon]^2,\ S(y)\ =\ \to -\frac{y}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on both sides separately

(co)-associativity on both sides

In[∘]:=
```
{(aΔ1→1,2 ~ B2 ~ aΔ2→2,3) ≡ (aΔ1→1,3 ~ B1 ~ aΔ1→1,2), (bΔ1→1,2 ~ B2 ~ bΔ2→2,3) ≡ (bΔ1→1,3 ~ B1 ~ bΔ1→1,2),
 (am1,2→1 ~ B1 ~ am1,3→1) ≡ (am2,3→2 ~ B2 ~ am1,2→1), (bm1,2→1 ~ B1 ~ bm1,3→1) ≡ (bm2,3→2 ~ B2 ~ bm1,2→1)}
```
Out[∘]=
```
{True, True, True, True}
```

Δ is an algebra morphism

In[∘]:=
```
{am1,2→1 ~ B1 ~ aΔ1→1,2 ≡ (aΔ1→1,3 aΔ2→2,4) ~ B1,2,3,4 ~ (am3,4→2 am1,2→1),
 bm1,2→1 ~ B1 ~ bΔ1→1,2 ≡ (bΔ1→1,3 bΔ2→2,4) ~ B1,2,3,4 ~ (bm3,4→2 bm1,2→1)}
```
Out[∘]=
```
{True, True}
```

S is convolution inverse of id

In[∘]:=
```
{(aΔ1→1,2 ~ B1 ~ aS1) ~ B1,2 ~ am1,2→1, (aΔ1→1,2 ~ B2 ~ aS2) ~ B1,2 ~ am1,2→1}
{(bΔ1→1,2 ~ B1 ~ bS1) ~ B1,2 ~ bm1,2→1, (bΔ1→1,2 ~ B2 ~ bS2) ~ B1,2 ~ bm1,2→1}
```
Out[∘]=
```
{𝔼[0, 0, 1 + O[ϵ]^2], 𝔼[0, 0, 1 + O[ϵ]^2]}
```
Out[∘]=
```
{𝔼[0, 0, 1 + O[ϵ]^2], 𝔼[0, 0, 1 + O[ϵ]^2]}
```

Si is the inverse of S

In[∘]:=
```
{aSi1 ~ B1 ~ aS1 ≡ 𝔼[a1 α1, x1 ξ1, 1], aS1 ~ B1 ~ aSi1 ≡ 𝔼[a1 α1, x1 ξ1, 1]}
{bSi1 ~ B1 ~ bS1 ≡ 𝔼[b1 β1, y1 η1, 1], bS1 ~ B1 ~ bSi1 ≡ 𝔼[b1 β1, y1 η1, 1]}
```
Out[∘]=
```
{True, True}
```
Out[∘]=
```
{True, True}
```

S is an algebra anti-(co)morphism

In[∘]:=
```
{am1,2→1 ~ B1 ~ aS1 ≡ (aS1 aS2) ~ B1,2 ~ am2,1→1, bm1,2→1 ~ B1 ~ bS1 ≡ (bS1 bS2) ~ B1,2 ~ bm2,1→1}
{aS1 ~ B1 ~ aΔ1→1,2 ≡ aΔ1→2,1 ~ B1,2 ~ (aS1 aS2), bS1 ~ B1 ~ bΔ1→1,2 ≡ bΔ1→2,1 ~ B1,2 ~ (bS1 bS2)}
```
Out[∘]=
```
{True, True}
```
Out[∘]=
```
{True, True}
```

Pairing

ybax

In[∘]:=
```
tPi_,j_ := 𝔼[βi αj, ηi ξj, 1 + 1/4 ϵ ηi^2 ξj^2]
```

In[∘]:=
```
qfac[k_, q_] := (1 - q)^-k QPochhammer[q, q, k] // FunctionExpand
qfe[k_] := Normal[Series[qfac[k, E^ρ], {ρ, 0, 1}]] /. {ρ → ϵ}
Table[𝔼[0, 0, y1^r b1^s a2^t x2^u] ~ B1,2 ~ tP1,2 ≡ 𝔼[0, 0, Kδr,u Kδs,t qfe[r] s!],
   {r, 0, 4}, {s, 0, 4}, {t, 0, 4}, {u, 0, 4}] // Flatten // Union
```
Out[∘]=
```
{True}
```

Pairing axioms

In[∘]:= $\{(bm_{1,2\to1}\,\mathbb{E}[\alpha_3 a_3, \xi_3 x_3, 1]) \sim B_{1,3}\sim tP_{1,3} \equiv (\mathbb{E}[\beta_1 b_1, \eta_1 y_1, 1]\,\mathbb{E}[\beta_2 b_2, \eta_2 y_2, 1]\, a\Delta_{3\to4,5}) \sim B_{1,4}\sim tP_{1,4}\sim B_{2,5}\sim tP_{2,5}, (b\Delta_{1\to1,2}\,\mathbb{E}[\alpha_3 a_3, \xi_3 x_3, 1]\,\mathbb{E}[\alpha_4 a_4, \xi_4 x_4, 1]) \sim B_{1,3}\sim tP_{1,3}\sim B_{2,4}\sim tP_{2,4} \equiv (\mathbb{E}[\beta_1 b_1, \eta_1 y_1, 1]\, am_{3,4\to3}) \sim B_{1,3}\sim tP_{1,3}\}$

Out[∘]= {True, True}

In[∘]:= $\{(bS_1\,\mathbb{E}[\alpha_2 a_2, \xi_2 x_2, 1]) \sim B_{1,2}\sim tP_{1,2} \equiv (\mathbb{E}[\beta_1 b_1, \eta_1 y_1, 1]\, aS_2) \sim B_{1,2}\sim tP_{1,2}, (bSi_1\,\mathbb{E}[\alpha_2 a_2, \xi_2 x_2, 1]) \sim B_{1,2}\sim tP_{1,2} \equiv (\mathbb{E}[\beta_1 b_1, \eta_1 y_1, 1]\, aSi_2) \sim B_{1,2}\sim tP_{1,2}\}$

Out[∘]= {True, True}

The Double

The double multiplication (should really bind the a's and b's separately)

ybax

In[∘]:= $\text{Block}[\{i, j, k\}, dm_{i\_,j\_\to k\_} = \text{Simplify} /@ \text{Expand} /@ ((\mathbb{E}[\beta_i b_i + \alpha_j a_j, \eta_i y_i + \xi_j x_j, 1]\,(a\Delta_{i\to h1,h2}\sim B_{h2}\sim a\Delta_{h2\to h2,h3})\,(b\Delta_{j\to t1,t2}\sim B_{t2}\sim b\Delta_{t2\to t2,t3}))\sim B_{h3}\sim aSi_{h3}\sim B_{t1,h3}\sim(tP_{t1,h3})\sim B_{t3,h1}\sim(tP_{t3,h1})\sim B_{h2,j,i,t2}\sim(am_{h2,j\to k}\,bm_{i,t2\to -k})\ /.\ \{u\_{-k} :> u_k\})]$

ybax

Out[∘]= $$\mathbb{E}\Big[a_k(\alpha_i+\alpha_j)+b_k(\beta_i+\beta_j),\ y_k\Big(\eta_i+\frac{\eta_j}{\mathcal{A}_i}\Big)-(-1+B_k)\,\eta_j\,\xi_i+x_k\Big(\frac{\xi_i}{\mathcal{A}_j}+\xi_j\Big),$$
$$1+\frac{1}{4\mathcal{A}_i\mathcal{A}_j}\Big(2 y_k \eta_j\big(2x_k\xi_i+\mathcal{A}_j(-2\beta_i+(1-3B_k)\eta_j\xi_i)\big)+\mathcal{A}_i\xi_i\big(x_k(-4\beta_j+2(1-3B_k)\eta_j\xi_i)+\mathcal{A}_j\eta_j(4a_kB_k+(1-4B_k+3B_k^2)\eta_j\xi_i)\big)\Big)\epsilon+O[\epsilon]^2\Big]$$

ybax

In[∘]:=
```
(*Deriving dS using dm*)
```
$\text{Block}[\{i\}, dS_{i\_} = ((bSi_1\, aS_2)\sim B_{1,2}\sim dm_{2,1\to i})\ /.\ \{z\_{1|2}\to z_i\}]$

ybax

Out[∘]= $$\mathbb{E}\Big[-a_i\alpha_i-b_i\beta_i,\ \frac{-y_i\mathcal{A}_i\eta_i-B_ix_i\mathcal{A}_i\xi_i+\mathcal{A}_i\eta_i\xi_i-B_i\mathcal{A}_i\eta_i\xi_i}{B_i},$$
$$1+\frac{1}{4B_i^2}\big(4B_iy_i\mathcal{A}_i\eta_i-4B_iy_i\mathcal{A}_i\beta_i\eta_i-2y_i^2\mathcal{A}_i^2\eta_i^2-4a_iB_i^2x_i\mathcal{A}_i\xi_i-4B_i^2x_i\mathcal{A}_i\beta_i\xi_i-4B_i\mathcal{A}_i\eta_i\xi_i+4a_iB_i\mathcal{A}_i\eta_i\xi_i+4B_i^2\mathcal{A}_i\eta_i\xi_i-4B_ix_iy_i\mathcal{A}_i^2\eta_i\xi_i+4B_i\mathcal{A}_i\beta_i\eta_i\xi_i-4B_i^2\mathcal{A}_i\beta_i\eta_i\xi_i+6y_i\mathcal{A}_i^2\eta_i^2\xi_i-2B_iy_i\mathcal{A}_i^2\eta_i^2\xi_i-2B_i^2x_i^2\mathcal{A}_i^2\xi_i^2+6B_ix_i\mathcal{A}_i^2\eta_i\xi_i^2-2B_i^2x_i\mathcal{A}_i^2\eta_i\xi_i^2-3\mathcal{A}_i^2\eta_i^2\xi_i^2+4B_i\mathcal{A}_i^2\eta_i^2\xi_i^2-B_i^2\mathcal{A}_i^2\eta_i^2\xi_i^2\big)\epsilon+O[\epsilon]^2\Big]$$

ybax

In[∘]:=
```
(*Deriving dΔ using dm*)
```
$\text{Block}[\{i, j, k\}, d\Delta_{i\_\to j\_,k\_} = (b\Delta_{i\to3,1}\, a\Delta_{i\to2,4})\sim B_{1,2,3,4}\sim(dm_{3,4\to k}\, dm_{1,2\to j})]$

ybax

Out[∘]= $$\mathbb{E}\Big[a_j\alpha_i+a_k\alpha_i+b_j\beta_i+b_k\beta_i,\ y_j\eta_i+B_jy_k\eta_i+x_j\xi_i+x_k\xi_i,\ 1+\frac{1}{2}\big(B_jy_jy_k\eta_i^2-2a_jx_k\xi_i+x_jx_k\xi_i^2\big)\epsilon+O[\epsilon]^2\Big]$$

First check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
{
  "[a,y]" → ((𝔼[0, 0, y₂ a₁] ~B₁,₂~ dm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, y₁ a₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧),
  "[b,x]" → ((𝔼[0, 0, x₂ b₁] ~B₁,₂~ dm₁,₂→₁)⟦3⟧ - (𝔼[0, 0, x₁ b₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧),
  "xy-qyx" → ((𝔼[0, 0, x₁ y₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧ - (1 + ϵ) (𝔼[0, 0, y₁ x₂] ~B₁,₂~ dm₁,₂→₁)⟦3⟧)
 } /. {z_1 → z} // Expand // Factor
{
 "Δ(a)" → ((𝔼[0, 0, a₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧),
 "Δ(x)" → ((𝔼[0, 0, x₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧),
 "Δ(b)" → ((𝔼[0, 0, b₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧),
 "Δ(y)" → ((𝔼[0, 0, y₁] ~B₁~ dΔ₁→₁,₂)⟦3⟧)
} // Simplify
{
 "S(a)" → ((𝔼[0, 0, a₁] ~B₁~ dS₁)⟦3⟧),
 "S(x)" → ((𝔼[0, 0, x₁] ~B₁~ dS₁)⟦3⟧),
 "S(b)" → ((𝔼[0, 0, b₁] ~B₁~ dS₁)⟦3⟧),
 "S(y)" → ((𝔼[0, 0, y₁] ~B₁~ dS₁)⟦3⟧)
} /. {z_1 → z} // Simplify
```

Out[∘]= $\{[a,y] \to -y + O[\epsilon]^2,\ [b,x] \to x\,\epsilon + O[\epsilon]^2,\ xy\text{-}qyx \to (1-B) + a\,B\,\epsilon + O[\epsilon]^2\}$

Out[∘]= $\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^2,\ \Delta(x) \to (x_1 + x_2) - a_1 x_2 \epsilon + O[\epsilon]^2,\ \Delta(b) \to (b_1 + b_2) + O[\epsilon]^2,\ \Delta(y) \to (y_1 + B_1 y_2) + O[\epsilon]^2\}$

Out[∘]= $\{S(a) \to -a + O[\epsilon]^2,\ S(x) \to -x - a\,x\,\epsilon + O[\epsilon]^2,\ S(b) \to -b + O[\epsilon]^2,\ S(y) \to -\frac{y}{B} + \frac{y\,\epsilon}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on double

(co)-associativity

In[∘]:=
```
{(dΔ₁→₁,₂ ~B₂~ dΔ₂→₂,₃) ≡ (dΔ₁→₁,₃ ~B₁~ dΔ₁→₁,₂),
 (dm₁,₂→₁ ~B₁~ dm₁,₃→₁) ≡ (dm₂,₃→₂ ~B₂~ dm₁,₂->₁)}
```

Out[∘]= {True, True}

Δ is an algebra morphism

In[∘]:=
```
dm₁,₂→₁ ~B₁~ dΔ₁→₁,₂ ≡ (dΔ₁→₁,₃ dΔ₂→₂,₄) ~B₁,₂,₃,₄~ (dm₃,₄→₂ dm₁,₂→₁)
```

Out[∘]= True

S is convolution inverse of id

In[∘]:=
```
{(dΔ₁→₁,₂ ~B₁~ dS₁) ~B₁,₂~ dm₁,₂→₁, (dΔ₁→₁,₂ ~B₂~ dS₂) ~B₁,₂~ dm₁,₂→₁}
```

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2],\ \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

S is a (co)-algebra anti-morphism

In[∘]:=
```
{dm₁,₂→₁ ~B₁~ dS₁ ≡ (dS₁ dS₂) ~B₁,₂~ dm₂,₁→₁, dS₁ ~B₁~ dΔ₁→₁,₂ ≡ dΔ₁→₂,₁ ~B₁,₂~ (dS₁ dS₂)} // Expand
```

Out[∘]= {True, True}

R-matrix

ybax

In[∘]:=

$$\mathbb{e}_{q\_,k\_}[x\_] := \mathbb{e}\wedge\left(\sum_{j=1}^{k+1}\frac{(1-q)^j x^j}{j\,(1-q^j)}\right)$$

$$R_{i\_,j\_} := \mathbb{E}\left[b_i\, a_j,\ y_i\, x_j,\ 1-\epsilon\,\frac{1}{4}\,y_i^2\, x_j^2 + O[\epsilon]^2\right]$$

(*First two terms in Faddeev-Quesne formula*)

In[∘]:= $\text{Series}[\mathbb{e}_{q,1}[z]\ /.\ \{z\to y_i\, x_j,\ q\to 1+\rho\},\ \{\rho,0,1\}]\ /.\ \{\rho\to\epsilon\}$

Out[∘]= $\mathbb{e}^{x_j y_i} - \frac{1}{4}\left(\mathbb{e}^{x_j y_i}\, x_j^2\, y_i^2\right)\epsilon + O[\epsilon]^2$

Quasi-triangular axiom 1:

In[∘]:= $R_{1,2}\sim B_1\sim d\Delta_{1\text{->}1,3} \equiv (R_{1,4}\, R_{3,2})\sim B_{2,4}\sim dm_{2,4\to 2}$

Out[∘]= True

Quasi-triangular axiom 2:

In[∘]:= $\left(\left(d\Delta_{1\text{->}1,2}\, R_{3,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right)\right) \equiv$
$\left(\left(d\Delta_{1\to 2,1}\, R_{3,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{3,1\to 1}\, dm_{4,2\to 2}\right)\right)$

Out[∘]= True

Reidemeister 3:

In[∘]:= $\left((R_{1,2}\, R_{4,3}\, R_{5,6})\sim B_{1,4}\sim dm_{1,4\to 1}\sim B_{2,5}\sim dm_{2,5\to 2}\sim B_{3,6}\sim dm_{3,6\to 3}\right) \equiv$
$\left((R_{1,6}\, R_{2,3}\, R_{4,5})\sim B_{1,4}\sim dm_{1,4\to 1}\sim B_{2,5}\sim dm_{2,5\to 2}\sim B_{3,6}\sim dm_{3,6\to 3}\right)$

Out[∘]= True

ybax

In[∘]:= $\text{Block}\left[\{i, j\},\ \overline{R}_{i\_,j\_} = \text{Expand}\ /@\ R_{i,j}\sim B_j\sim dS_j\right]$

ybax

Out[∘]= $\mathbb{E}\left[-a_j\, b_i,\ -\frac{x_j\, y_i}{B_i},\ 1+\frac{\left(-4\, a_j\, B_i\, x_j\, y_i - 3\, x_j^2\, y_i^2\right)\epsilon}{4\, B_i^2} + O[\epsilon]^2\right]$

Reidemeister 2

In[∘]:= $\left\{\left(\overline{R}_{1,2}\, R_{3,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right),\ \left(R_{1,2}\, \overline{R}_{3,4}\right)\sim B_{1,2,3,4}\sim\left(dm_{1,3\to 1}\, dm_{2,4\to 2}\right)\right\}$

Out[∘]= $\left\{\mathbb{E}\left[0, 0, 1+O[\epsilon]^2\right],\ \mathbb{E}\left[0, 0, 1+O[\epsilon]^2\right]\right\}$

Deriving the Drinfeld element u and its inverse ui

ybax

In[∘]:=
```
Block[{i}, {
  u_i_ = R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i},
  ui_i_ := R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→i}
}]
```

ybax

Out[∘]= $\left\{\mathbb{E}\left[-a_i b_i,\ -\frac{x_i y_i}{B_i},\ B_i+\frac{\left(-4 a_i B_i^2-4 B_i x_i y_i-4 a_i B_i x_i y_i-3 x_i^2 y_i^2\right)\epsilon}{4 B_i}+O[\epsilon]^2\right],\ \text{Null}\right\}$

u and ui are inverses

In[∘]:= $\left(u_1\, ui_2\right) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0,\ 0,\ 1+O[\epsilon]^2\right]$

The ribbon element v satisfies $v^2 = S(u)\, u$. The spinner C=uv^{-1}.
It is convenient to compute $z = S(u)\, u^{-1}$ which is something easy.

In[∘]:= $\left(\left(u_1 \sim B_1 \sim dS_1\right) ui_2\right) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0,\ 0,\ \frac{1}{B_1}+\frac{a_1\,\epsilon}{B_1}+O[\epsilon]^2\right]$

(* Needs fixing! *) So in our case S(u) = u z so S(u)u = $u^2 z$ and v = u$z^{\frac{1}{2}}$ and finally C = uv^{-1} = $z^{-\frac{1}{2}} = e^{t_1/2}(1-\epsilon a_1)$.

ybax

In[∘]:=
```
Block[{i},
 {CC_i_ = 𝔼[0, 0, B_i^(1/2) e^(-ϵ a_i/2) + O[ϵ]^2],
  CC̄_i_ = 𝔼[0, 0, B_i^(-1/2) e^(ϵ a_i/2) + O[ϵ]^2]
 }]
```

ybax

Out[∘]= $\left\{\mathbb{E}\left[0,\ 0,\ \sqrt{B_i}-\frac{1}{2}\left(a_i\sqrt{B_i}\right)\epsilon+O[\epsilon]^2\right],\ \mathbb{E}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}}+\frac{a_i\,\epsilon}{2\sqrt{B_i}}+O[\epsilon]^2\right]\right\}$

ybax

In[∘]:=
```
Block[{i, j}, {
  Kink_i_ = (R_{1,3} CC̄_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i},
  K̄ink_j_ = (R̄_{1,3} CC_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j}
}]
```

ybax

Out[∘]= $\left\{\mathbb{E}\left[a_i b_i,\ x_i y_i,\ \frac{1}{\sqrt{B_i}}+\frac{\left(2 a_i-x_i^2 y_i^2\right)\epsilon}{4\sqrt{B_i}}+O[\epsilon]^2\right],\right.$
$\left.\mathbb{E}\left[-a_j b_j,\ -\frac{x_j y_j}{B_j},\ \sqrt{B_j}+\frac{\left(-2 a_j B_j^2-4 a_j B_j x_j y_j-3 x_j^2 y_j^2\right)\epsilon}{4 B_j^{3/2}}+O[\epsilon]^2\right]\right\}$

In[∘]:=
$$k2 = \left(R_{3,1}\, CC_2\right) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to i} \;/.\; e \to E;$$
$$k4 = \left(\overline{R}_{3,1}\, \overline{CC}_2\right) \sim B_{1,2} \sim dm_{1,2\to 1} \sim B_{1,3} \sim dm_{1,3\to j} \;/.\; e \to E;$$
$$\text{Simplify}@\left\{\text{Kink}_i \equiv k2,\ \overline{\text{Kink}}_j \equiv k4,\ \left(\text{Kink}_i\, \overline{\text{Kink}}_j\right) \sim B_{i,j} \sim dm_{i,j\to 1}\right\}$$

Out[∘]= $\left\{\text{True},\ \text{True},\ \mathbb{E}\left[0,\ 0,\ 1+O[\epsilon]^2\right]\right\}$

Reidemeister 2:

In[∘]:= $\left(R_{1,2}\, \overline{R}_{3,4}\right) \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{2,4} \sim dm_{2,4\to 2}$

Out[∘]= $\mathbb{E}\left[0,\ 0,\ 1+O[\epsilon]^2\right]$

Cyclic Reidemeister 2:

In[∘]:= $\left(R_{1,4}\, \overline{R}_{5,2}\, \overline{CC}_3\right) \sim B_{2,4} \sim dm_{2,4\to 2} \sim B_{1,3} \sim dm_{1,3\to 1} \sim B_{1,5} \sim dm_{1,5\to 1} \equiv \overline{CC}_1$

Out[∘]= True

Trefoil

In[∘]:=
```
Timing[
 Z = R1,5 R6,2 R3,7 CC̄4 Kink̄8 Kink̄9 Kink̄10;
 Do[Print["doing ", r]; Z = Z~B1,r~dm1,r→1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```
$$Z = R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC}_4\, \overline{\text{Kink}}_8\, \overline{\text{Kink}}_9\, \overline{\text{Kink}}_{10};$$
$$\text{Do}[\text{Print}[\text{"doing "}, r];\ Z = Z \sim B_{1,r} \sim dm_{1,r\to 1},\ \{r, 2, 10\}];$$

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[∘]=
$$\left\{106.188,\ \frac{B_1}{1-B_1+B_1^2} + \left(B_1\left(-B_1+2B_1^2+2B_1^4+a_1\left(-1+B_1-B_1^3+B_1^4\right)-2x_1y_1-B_1^3\left(3+2x_1y_1\right)\right)\epsilon\right) \Big/ \left(1-B_1+B_1^2\right)^3 + O[\epsilon]^2\right\}$$

---

ybax

In[∘]:=
$$\text{b2t}_{i\_} := \mathbb{E}\left[\alpha_i a_i - \beta_i t_i,\ \xi_i x_i + \eta_i y_i,\ 1 + \epsilon\, \beta_i a_i + O[\epsilon]^2\right]$$
$$\text{t2b}_{i\_} := \mathbb{E}\left[\alpha_i a_i - \tau_i b_i,\ \xi_i x_i + \eta_i y_i,\ 1 + \epsilon\, \tau_i a_i + O[\epsilon]^2\right]$$

In[]:= $R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC_4}\, \overline{Kink_8}\, \overline{Kink_9}\, \overline{Kink_{10}}$

Out[]= $$\mathbb{E}\Big[a_5 b_1 + a_7 b_3 + a_2 b_6 - a_8 b_8 - a_9 b_9 - a_{10} b_{10},\; x_5 y_1 + x_7 y_3 + x_2 y_6 - \frac{x_8 y_8}{B_8} - \frac{x_9 y_9}{B_9} - \frac{x_{10} y_{10}}{B_{10}},\; \frac{\sqrt{B_8}\,\sqrt{B_9}\,\sqrt{B_{10}}}{\sqrt{B_4}} + \left(\sqrt{B_{10}}\left(\sqrt{B_9}\left(\sqrt{B_8}\left(\frac{a_4}{2\sqrt{B_4}} - \frac{x_5^2 y_1^2}{4\sqrt{B_4}} - \frac{x_7^2 y_3^2}{4\sqrt{B_4}} - \frac{x_2^2 y_6^2}{4\sqrt{B_4}}\right) + \frac{-2 a_8 B_8^2 - 4 a_8 B_8 x_8 y_8 - 3 x_8^2 y_8^2}{4\sqrt{B_4}\, B_8^{3/2}}\right) + \frac{\sqrt{B_8}\left(-2 a_9 B_9^2 - 4 a_9 B_9 x_9 y_9 - 3 x_9^2 y_9^2\right)}{4\sqrt{B_4}\, B_9^{3/2}}\right) + \frac{1}{4\sqrt{B_4}\, B_{10}^{3/2}}\sqrt{B_8}\,\sqrt{B_9}\left(-2 a_{10} B_{10}^2 - 4 a_{10} B_{10} x_{10} y_{10} - 3 x_{10}^2 y_{10}^2\right)\right)\epsilon + O[\epsilon]^2\Big]$$

In[]:= $\left(R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC_4}\, \overline{Kink_8}\, \overline{Kink_9}\, \overline{Kink_{10}}\right) \sim B_{Range[10]} \sim$ Product[$b2t_i$, {i, 10}]

Out[]= $$\mathbb{E}\Big[-a_5 t_1 - a_7 t_3 - a_2 t_6 + a_8 t_8 + a_9 t_9 + a_{10} t_{10},\; \frac{1}{T_8 T_9 T_{10}}\left(T_8 T_9 T_{10} x_5 y_1 + T_8 T_9 T_{10} x_7 y_3 + T_8 T_9 T_{10} x_2 y_6 - T_9 T_{10} x_8 y_8 - T_8 T_{10} x_9 y_9 - T_8 T_9 x_{10} y_{10}\right),\; \frac{\sqrt{T_8}\,\sqrt{T_9}\,\sqrt{T_{10}}}{\sqrt{T_4}} + \frac{1}{4\sqrt{T_4}\, T_8^{3/2} T_9^{3/2} T_{10}^{3/2}}\left(4 a_4 T_8^2 T_9^2 T_{10}^2 + 4 a_1 a_5 T_8^2 T_9^2 T_{10}^2 + 4 a_2 a_6 T_8^2 T_9^2 T_{10}^2 + 4 a_3 a_7 T_8^2 T_9^2 T_{10}^2 - 4 a_8 T_8^2 T_9^2 T_{10}^2 - 4 a_8^2 T_8^2 T_9^2 T_{10}^2 - 4 a_9 T_8^2 T_9^2 T_{10}^2 - 4 a_9^2 T_8^2 T_9^2 T_{10}^2 - 4 a_{10} T_8^2 T_9^2 T_{10}^2 - 4 a_{10}^2 T_8^2 T_9^2 T_{10}^2 - T_8^2 T_9^2 T_{10}^2 x_5^2 y_1^2 - T_8^2 T_9^2 T_{10}^2 x_7^2 y_3^2 - T_8^2 T_9^2 T_{10}^2 x_2^2 y_6^2 - 8 a_8 T_8 T_9^2 T_{10}^2 x_8 y_8 - 3 T_9^2 T_{10}^2 x_8^2 y_8^2 - 8 a_9 T_8^2 T_9 T_{10}^2 x_9 y_9 - 3 T_8^2 T_{10}^2 x_9^2 y_9^2 - 8 a_{10} T_8^2 T_9^2 T_{10} x_{10} y_{10} - 3 T_8^2 T_9^2 x_{10}^2 y_{10}^2\right)\epsilon + O[\epsilon]^2\Big]$$

In[]:= Z = $\left(\left(\left(R_{1,5}\, R_{6,2}\, R_{3,7}\, \overline{CC_4}\, \overline{Kink_8}\, \overline{Kink_9}\, \overline{Kink_{10}}\right) \sim B_{Range[10]} \sim \text{Product}[b2t_i, \{i, 10\}]\right) \text{ /. } T\_ \to T_1\right) \sim B_{Range[10]} \sim$ Product[$t2b_i$, {i, 10}]

Out[]= $$\mathbb{E}\Big[a_5 b_1 + a_7 b_3 + a_2 b_6 - a_8 b_8 - a_9 b_9 - a_{10} b_{10},\; \frac{1}{B_1}\left(B_1 x_5 y_1 + B_1 x_7 y_3 + B_1 x_2 y_6 - x_8 y_8 - x_9 y_9 - x_{10} y_{10}\right),\; B_1 + \frac{1}{4 B_1}\left(4 a_1 B_1^2 + 4 a_4 B_1^2 - 4 a_8 B_1^2 - 4 a_9 B_1^2 - 4 a_{10} B_1^2 - B_1^2 x_5^2 y_1^2 - B_1^2 x_7^2 y_3^2 - B_1^2 x_2^2 y_6^2 + 4 a_1 B_1 x_8 y_8 - 8 a_8 B_1 x_8 y_8 - 3 x_8^2 y_8^2 + 4 a_1 B_1 x_9 y_9 - 8 a_9 B_1 x_9 y_9 - 3 x_9^2 y_9^2 + 4 a_1 B_1 x_{10} y_{10} - 8 a_{10} B_1 x_{10} y_{10} - 3 x_{10}^2 y_{10}^2\right)\epsilon + O[\epsilon]^2\Big]$$

In[]:=
```
Timing[
 Do[Print["doing ", r]; Z = Z~B1,r~dm1,r→1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[∘]= $\left\{6.23438, \frac{B_1}{1-B_1+B_1^2} + \left(B_1 \left(-B_1 + 2 B_1^2 + 2 B_1^4 + a_1 \left(-1 + B_1 - B_1^3 + B_1^4\right) - 2 x_1 y_1 - B_1^3 \left(3 + 2 x_1 y_1\right)\right) \epsilon\right) \Big/ \left(1 - B_1 + B_1^2\right)^3 + O[\epsilon]^2\right\}$

In[∘]:=
```
Timing[
 Z = R1,5 R6,2 R3,7 CC4‾ Kink8‾ Kink9‾ Kink10‾ /. B_ → B1;
 Do[Print["doing ", r]; Z = Z~B1,r~dm1,r→1 /. B_ → B1, {r, 2, 10}];
 Simplify@Z[[3]] ]
```

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[∘]= $\left\{6.26563, \frac{B_1}{1-B_1+B_1^2} + \frac{1}{2 B_1 \left(1-B_1+B_1^2\right)^3} \left(2 a_1 B_1^2 \left(-1 + B_1 - B_1^3 + B_1^4\right) - 6 x_1^2 y_1^2 + 4 B_1^7 x_1^2 y_1^2 - 2 B_1^8 x_1^2 y_1^2 + B_1^2 x_1 y_1 \left(5 - 6 x_1 y_1\right) + 3 B_1 x_1 y_1 \left(-1 + 2 x_1 y_1\right) + B_1^6 \left(3 + 3 x_1 y_1 - 6 x_1^2 y_1^2\right) - B_1^5 \left(4 + 13 x_1 y_1 + 2 x_1^2 y_1^2\right) + B_1^4 \left(2 + 15 x_1 y_1 + 4 x_1^2 y_1^2\right) - B_1^3 \left(1 + 15 x_1 y_1 + 6 x_1^2 y_1^2\right)\right) \epsilon + O[\epsilon]^2\right\}$